HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Optometry I (General)



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⋇. When you see ⋇ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAMPLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE"	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [.] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

* Requires one memory module.

ANISEIKONIA

(requires one memory module)

This program calculates the spectacle magnification of the R_x of interest. It is calculated for both principle meridians of both eyes and is listed as magnification and axis. Inputs are readings from an A.O. Space Eikonometer which is converted into magnification needed by the patient tested.

Equations:

$$x = (x180°) - (x90°)$$

$$y = 3.5\delta$$

$$f = \sqrt{x^2 + y^2}$$

$$M\phi = \tfrac{1}{2}[(x90°) + (x180°) + f]$$

$$M\phi \pm 90° = \tfrac{1}{2}[(x90°) + (x180°) - f]$$

$$M = \left(\frac{1}{1-D_1 t/n}\right)\left(\frac{1}{1 - Vh}\right)$$

Where:

x = x component; y = y component;

$x90°$ = Eikonometer value in %; $x180°$ = Eikonometer value in %;

δ = declination angle, another Eikonometer value, in degrees;

f = hypotenuse from x and y;

$M\phi$ = magnification needed in meridian ϕ, in %;

$M\phi \pm 90$ = magnification needed in meridian $\phi \pm 90$, in %;

M = magnification; D_1 = Dioptric power of front surface of lens;

t = Center thickness of lens in meters;

n = Index of refraction of lens material;

V = Back vertex power of same meridian as D_1 in Diopters;

h = Distance from corneal apex to rear vertex of lens in meters.

NOTE:

h is calculated from frame eye size; distance to edge of front surface of lens from corneal apex; n; D in 180th meridian of lens; t of lens by using SAG formula. This automatically corrects h as D_1 is changed whether eye wire distance is changed or not.

Results may differ slightly from those obtained by use of the table, since the calculator does not round them off, hence its answers are more accurate.

Reference:

A.O. Space Eikonometer instructions for graphical solutions, Pacific University Aniseikonia Manual.

The magnification needed will be correct for all values obtained with the space Eikonometer, including those beyond the range of the table supplied for use with the instrument. Values from the instrument indicating left eye are entered as minus; those indicating right eye are entered as plus. Similarly plus output means mangification before right eye, and minus output means magnification before left eye.

Example:

If x90° = 1.2% right eye

x180° = 0.4% left eye

δ = -0.8°

then what magnification is needed by this patient?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 027	
[XEQ] [ALPHA] ANSK [ALPHA]	SPEC. MAG.
	%RT↑ %LFT↑ Δ
1.2 [ENTER↑] 0.4 [CHS] [ENTER↑]	
0.8 [CHS] [R/S]	2.01 %
[R/S]	120 DEG.
[R/S]	-1.21 %
[R/S]	30. DEG.

To calculate spectacle magnifications, the Rx and other measurements are:

Rx (OD): +1.00 - 2.00 x 30

Rx (OS): +2.00 - 1.00 x 120

Eye size = 50mm

n = 1.523

Lens edge distance OU = 14mm

Dc = Dc±90° = +6.00 D. OD

t_{OD} = 2.8mm

Dc = Dc±90° = +6.50 D. OS

t_{OS} = 3.4mm

Note: Dc = D in meridian of cylinder axis

Dc 90^{o} = D 90^{o} from cylinder axis.

The meridians must not be oblique to the cylinder axis as the program will give a wrong answer.

[R/S]	KEY RX, OD
	SPH↑ CYL↑ AX
1 [ENTER] 2 [CHS] [ENTER↑]	
30 [R/S]	KEY RX, OS
	SPH↑ CYL↑ AX
2 [ENTER↑] 1 [CHS] [ENTER↑]	
120 [R/S]	KEY OU DATA
	FS↑ N↑ DIST
50 [ENTER↑] 1.523 [ENTER↑] 14 [R/S]	KEY OD DATA
	PCA↑ PFS↑ T
6 [ENTER↑] [ENTER↑] 2.8 [R/S]	KEY OS DATA
	PCA↑ PFS↑ T
6.5 [ENTER↑] [ENTER↑] 3.4 [R/S]	1.0264 %
[R/S]	120 DEG.
[R/S]	0.9963 %
[R/S]	30. DEG.
[R/S]	1.0452 %
[R/S]	30. DEG.
[R/S]	1.0297 %
[R/S]	120. DEG.

User Instructions

SIZE: 027

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] ANSK	SPECT. MAG.
				%RT↑ %LFT↑ δ
3.	Enter x90° value.	x90° (%)	[ENTER↑]	
	Enter x180° value.	x180° (%)	[ENTER↑]	
	Enter δ value and calculate magnification			
	needed.	δ (DEG)	[R/S]	( )%
	Note: Use minus values for left eye,			
	plus values for right eye.			
			[R/S]	( )DEG.
			[R/S]	( )%
			[R/S]	( ) DEG.
4.	For spectacle magnification:		[R/S]	KEY RX OD
				SPH↑ CYL↑ AX
5.	Enter spectacle Rx, OD in minus cylinder			
	form:			
	Enter SPH.	SPH (D)	[ENTER↑]	
	Enter CYL.	CYL (D)	[ENTER↑]	
	Enter AXIS.	AXIS (DEG)	[R/S]	KEY RX OS
				SPH↑ CYL↑ AX
6.	Now enter spectacle Rx, OS in minus cylin-			
	der form:			
	Enter SPH.	SPH (D)	[ENTER↑]	
	Enter CYL.	CYL (D)	[ENTER↑]	
	Enter AXIS.	AXIS (DEG)	[R/S]	KEY OU DATA
				FS↑ N↑ DIST
	(cont'd)			

User Instructions

SIZE: 027

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
7.	Now enter OU data:			
	Enter spectacle frame eye size.	FS (mm)	[ENTER↑]	
	Enter n of lenses.	n	[ENTER↑]	
	Enter corneal apex to edge of front surface			
	of lens distance in 180th meridian.	DIST (mm)	[R/S]	KEY OD DATA
				PCA↑ PFS↑ T
8.	Now enter OD data:			
	Enter power of front surface in meridian			
	of cylinder axis.	PCA (D)	[ENTER↑]	
	Enter power of front surface in meridian			
	90° from CYL.	PFS (D)	[ENTER↑]	
	Enter center thickness of lens.	T_{OD} (mm)	[R/S]	KEY OS DATA
				PCA↑ PFS↑ T
9.	Now enter OS data.			
	Enter data in the format as in step 8, ex-	PCA (D)	[ENTER↑]	
	cept use values for OS lens and calculate	PFS (D)	[ENTER↑]	
	induced magnification.	T_{OS} (mm)	[R/S]	( )%
			[R/S]	( )DEG.
			[R/S]	( )%
			[R/S]	( )DEG.
			[R/S]	( )%
			[R/S]	( )DEG.
			[R/S]	( )%
			[R/S]	( )DEG.

Program Listings

```
 01*LBL "ANS
K"
 02 CF 21
 03 " SPECT.
 MAG."
 04 AVIEW
 05 PSE
 06 SF 21
 07 "%RT↑ %L
FT↑ ∠"
 08 PROMPT
 09 3.5
 10 *
 11 STO 00
 12 RDN
 13 STO 01
 14 RDN
 15 STO 02
 16 R↑
 17 X<>Y
 18 -
 19 STO 03
 20 X↑2
 21 RCL 00
 22 X↑2
 23 +
 24 SQRT
 25 STO 04
 26 RCL 00
 27 X=0?
 28 GTO 01
 29 RCL 03
 30 X=0?
 31 GTO 02
 32 /
 33 STO 05
 34 X>0?
 35 GTO 03
 36 RCL 00
 37 X>0?
 38 GTO 04
 39 360
 40 GTO 05
 41*LBL 03
 42 RCL 00
 43 X>0?
 44 GTO 05
 45*LBL 04
 46 180
 47*LBL 05
 48 RCL 05
```

Calculate magnitude needed

Correct axis

Calculate axis

```
 49 ATAN
 50 +
 51 2
 52 /
 53*LBL 06
 54 STO 06
 55 ADV
 56 90
 57 X<=Y?
 58 -
 59 X>Y?
 60 +
 61 STO 07
 62 RCL 01
 63 RCL 02
 64 +
 65 RCL 04
 66 +
 67 2
 68 /
 69 STO 08
 70 XEQ 11
 71 RCL 06
 72 XEQ 12
 73 RCL 01
 74 RCL 02
 75 +
 76 RCL 04
 77 -
 78 2
 79 /
 80 STO 09
 81 XEQ 11
 82 RCL 07
 83 XEQ 12
 84 GTO 13
 85*LBL 12
 86 FIX 0
 87 CLA
 88 ARCL X
 89 "⊢ DEG."
 90 AVIEW
 91 RTN
 92*LBL 11
 93 FIX 2
 94 CLA
 95 ARCL X
 96 "⊢ %"
 97 AVIEW
 98 RTN
 99*LBL 13
```

Calculate Mø and Mø ± 90

Program Listings

```
100 "KEY RX,
 OD"
101 CF 21
102 AVIEW
103 PSE
104 XEQ 09
105 RCL Z
106 STO 00
107 RCL Z
108 STO 01
109 RCL Z
110 STO 02
111 "KEY RX,
 OS"
112 AVIEW
113 PSE
114 XEQ 09
115 RCL Z
116 STO 03
117 RCL Z
118 STO 04
119 RCL Z
120 STO 05
121 "KEY OU
DATA"
122 AVIEW
123 PSE
124 "FS↑ N↑
DIST"
125 PROMPT
126 RCL Z
127 1 E3
128 /
129 STO 10
130 RCL Z
131 STO 11
132 RCL Z
133 1 E3
134 /
135 STO 12
136 "KEY OD
DATA"
137 AVIEW
138 PSE
139 XEQ 00
140 RCL Z
141 STO 13
142 RCL Z
143 STO 14
144 RCL Z
145 1 E3
```

Input RX, OD

Input RX, OS

Input OU data

Input OD data

```
146 /
147 STO 15
148 "KEY OS
DATA"
149 AVIEW
150 PSE
151 XEQ 00
152 RCL Z
153 STO 17
154 RCL Z
155 STO 18
156 RCL Z
157 1 E3
158 /
159 STO 19
160 SF 21
161 GTO 20
162◆LBL 09
163 "SPH↑ CY
L↑ AX"
164 PROMPT
165 RTN
166◆LBL 00
167 "PCA↑ PF
S↑ T"
168 PROMPT
169 RTN
170◆LBL 20
171 " %"
172 ASTO 16
173 " DEG."
174 ASTO 26
175 RCL 13
176 RCL 14
177 X>Y?
178 GTO 03
179 -
180 180
181 STO 24
182 RCL 02
183 -
184 SIN
185 X↑2
186 *
187 RCL 13
188 X<>Y
189 -
190 STO 25
191 XEQ 04
192 GTO 02
193◆LBL 03
```

Input OS data

Calculate spectacle magnifications

Program Listings

```
194 X<>Y
195 -
196 RCL 24
197 RCL 12
198 -
199 SIN
200 X↑2
201 *
202 RCL 03
203 +
204 STO 25
205 XEQ 04
206◆LBL 02
207 RCL 15
208 -
209 STO 25
210 1
211 RCL 13
212 RCL 15
213 *
214 RCL 11
215 /
216 -
217 1/X
218 1
219 RCL 00
220 RCL 25
221 *
222 -
223 1/X
224 *
225 STO 20
226 XEQ 11
227 90
228 RCL 02
229 X<=Y?
230 GTO 05
231 X<>Y
232 -
233 XEQ 12
234 GTO 06
235◆LBL 05
236 +
237 XEQ 12
238◆LBL 06
239 1
240 RCL 00
241 RCL 01
242 +
243 RCL 25
244 *
```

```
245 -
246 1/X
247 1
248 RCL 14
249 RCL 15
250 *
251 RCL 11
252 /
253 -
254 1/X
255 *
256 STO 21
257 XEQ 11
258 RCL 02
259 XEQ 12
260 RCL 17
261 RCL 18
262 X>Y?
263 GTO 07
264 -
265 RCL 24
266 RCL 05
267 -
268 SIN
269 X↑2
270 *
271 RCL 17
272 X<>Y
273 -
274 STO 25
275 XEQ 04
276 GTO 08
277◆LBL 07
278 X<>Y
279 -
280 RCL 24
281 RCL 05
282 -
283 SIN
284 X↑2
285 *
286 RCL 17
287 +
288 STO 25
289 XEQ 04
290◆LBL 08
291 RCL 19
292 -
293 STO 25
294 1
295 RCL 17
```

Program Listings

```
296 RCL 19
297 *
298 RCL 11
299 /
300 -
301 1/X
302 1
303 RCL 03
304 RCL 25
305 *
306 -
307 1/X
308 *
309 STO 22
310 XEQ 11
311 90
312 RCL 05
313 X<=Y?
314 GTO 09
315 X<>Y
316 -
317 XEQ 12
318 GTO 13
319◆LBL 09
320 +
321 XEQ 12
322◆LBL 13
323 1
324 RCL 03
325 RCL 04
326 +
327 RCL 25
328 *
329 -
330 1/X
331 1
332 RCL 18
333 RCL 19
334 *
335 RCL 11
336 /
337 -
338 1/X
339 *
340 STO 23
341 XEQ 11
342 RCL 05
343 XEQ 12
344 GTO 10
345◆LBL 04
346 RCL 11
347 1
348 -
349 RCL 25
350 /
351 STO 25
352 X↑2
353 RCL 10
354 2
355 /
356 X↑2
357 -
358 SQRT
359 RCL 25
360 X<>Y
361 -
362 RCL 12
363 +
364 RTN
365◆LBL 11
366 FIX 4
367 CLA
368 ARCL X
369 ARCL 16
370 AVIEW
371 RTN
372◆LBL 12
373 FIX 0
374 CLA
375 ARCL X
376 ARCL 26
377 AVIEW
378◆LBL 10
379 ADV
380 RTN
381◆LBL 01
382 90
383 GTO 06
384◆LBL 02
385 45
386 GTO 06
387 END
```

00

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Scratch	50	
	Scratch		
	Scratch		
	Scratch		
	Scratch		
05	Scratch	55	
	ANS 2		
	ANS 4		
	ANS 1		
	ANS 3		
10	Scratch	60	
	Scratch		
	Scratch		
	Scratch		
	Scratch		
15	Scratch	65	
	"%"		
	Scratch		
	Scratch		
	Scratch		
20	ANS 5	70	
	ANS 7		
	ANS 9		
	ANS 11		
	Scratch		
25	Scratch	75	
	"DEG"		
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 027 TOT. REG. 124 USER MODE

ENG ______ FIX ______ SCI ______ ON ______ OFF ______

DEG ______ RAD ______ GRAD ______

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CROSSED PRISM RESULTANT

This program calculates the resultant power, and the base/apex meridian from two crossed prisms.

Equations:

$$R = \sqrt{A^2 + B^2 + 2AB \cos a}$$

$$\sin b = \frac{B}{R} \sin a$$

where:

A = Power of prism 1

B = Power of prism 2

R = Power of Resultant

a = angle between base/apex lines of 1 and 2

b = angle between base/apex lines 1 and R

Note: Prism 1 is taken to be the one whose base/apex meridian is smaller numerically.

Reference:

Clinical Refraction by Irvine M. Borish

Example:

Calculate the resultant power and its base/apex meridian of the following two prisms.

Prism 1 = 2Δ at 10°

Prism 2 = 3Δ at 55°

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] PRISM [ALPHA]	PWR1↑ MER1
2 [ENTER] 10 [R/S]	PWR2↑ MER2
3 [ENTER↑] 55 [R/S]	4.64 Prism D.
[R/S]	37.2 DEG.

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] PRISM	PWR1↑ MER1
3.	Key Prism 1 power.	A	[ENTER↑]	
4.	Key Prism 1 base/apex meridian.	(deg)	[R/S]	PWR2↑ MER2
5.	Key Prism 2 power.	B	[ENTER↑]	
6.	Key Prism 2 base/apex meridian.	(deg)	[R/S]	( ) PRISM D.
			[R/S]	( ) DEG.
7.	For new problem, press [R/S] and go to			
	step 3.		[R/S]	PWR1↑ MER1

Program Listings

```
 01◆LBL "PRI
SM"
 02◆LBL 01
 03 "PWR1↑ M
ER1"
 04 PROMPT
 05 STO 01
 06 RDN
 07 STO 00
 08 "PWR2↑ M
ER2"
 09 PROMPT
 10 STO 03
 11 RDN
 12 STO 02
 13 RCL 00
 14 X↑2
 15 RCL 02
 16 X↑2
 17 +
 18 RCL 00
 19 RCL 02
 20 *
 21 2
 22 *
 23 RCL 03
 24 RCL 01
 25 -
 26 STO 04
 27 COS
 28 *
 29 +
 30 SQRT
 31 FIX 2
 32 CLA
 33 ARCL X
 34 "⊢ PRISM
D."
 35 AVIEW
 36 STOP
 37 RCL 02
 38 X<>Y
 39 /
 40 RCL 04
 41 SIN
 42 *
 43 ASIN
 44 RCL 01
 45 +
 46 FIX 1
 47 CLA
 48 ARCL X
 49 "⊢ DEG."
 50 AVIEW
 51 STOP
 52 GTO 01
 53 .END.
```

Store prism 1 data

Store prism 2 data

Calculate R

Output R

Calculate angle

Output angle

60 70 80 90 00

 # REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	P1 Power	50	
	P1 b/a meridian		
	P2 Power		
	P2 b/a meridian		
	Scratch		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 19 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

OBLIQUE CYLINDER SUMS

This program will calculate the sum of two sphero-cylinder lenses, regardless of their axes. The cylinders are added in a graphical method by doubling their axes, converting to rectangular coordinates, summing coordinates, converting coordinates back to polar, and halving the axis to obtain the cylinder resultant. The new sphere is found from:

Equations:

$$S_1 + S_2 + \frac{(C_1 + C_2 - R)}{2} = S_3$$

where:

S_1 = Sphere power of lens 1

S_2 = Sphere power of lens 2

S_3 = Resultant sphere power

C_1 = Cylinder power of lens 1

C_2 = Cylinder power of lens 2

R = Resultant cylinder power

Remarks:

1. Lens 1 and Lens 2 data may be in plus or minus cylinder form.
2. Answers are in minus cylinder form.

Reference:

Cylinder resultant obtained from Mathematical Model of Humphrey Plot Polar Graphical Method. New sphere power calculated as in Clinical Refraction by Irvin M. Borish.

Example:

Calculate the sum of the following two lenses.

Lens 1: 4.50 - 1.25 x 100

Lens 2: -1.00 - 1.75 x 45

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] CSUM [ALPHA]	LENS 1
	SPH↑ CYL↑ AX
4.5 [ENTER↑] 1.25 [CHS] [ENTER↑]	
100 [R/S]	LENS 2
	SPH↑ CYL↑ AX
1 [CHS] [ENTER↑] 1.75 [CHS] [ENTER↑]	
45 [R/S]	2.88 SPH SUM
[R/S]	-1.77 CYL SUM
[R/S]	66. AX SUM

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] CSUM	LENS 1
				SPH↑ CYL↑ AX
3.	Key lens 1 sphere.	S_1	[ENTER↑]	
4.	Key lens 1 cylinder.	C_1	[ENTER↑]	
5.	Key lens 1 axis.	$axis_1$	[R/S]	LENS 2
		S_2		SPH↑ CYL↑ AX
6.	Key lens 2 shpere.	C_2	[ENTER↑]	
7.	Key lens 2 cylinder.	$axis_2$	[ENTER↑]	
8.	Key lens 2 axis.		[R/S]	( ) SPH SUM
			[R/S]	( ) CYL SUM
			[R/S]	( ) AX SUM
9.	For new problem, press [R/S] and go to			
	step 3.		[R/S]	LENS 1
				SPH↑ CYL↑ AX
	Note: A third lens may be summed by keying			
	the resultant data as the new lens 1 and			
	the third lens as lens 2.			

Program Listings

```
 01◆LBL "CSU
M"
 02 FIX 2
 03◆LBL 03
 04 "LENS 1"
 05 AVIEW
 06 PSE
 07 "SPH↑ CY
L↑ AX"
 08 PROMPT
 09 STO 02
 10 RDN
 11 STO 01
 12 RDN
 13 STO 00
 14 "LENS 2"
 15 AVIEW
 16 PSE
 17 "SPH↑ CY
L↑ AX"
 18 PROMPT
 19 STO 05
 20 RDN
 21 STO 04
 22 RDN
 23 STO 03
 24 RCL 02
 25 2
 26 *
 27 RCL 01
 28 P-R
 29 STO 06
 30 RDN
 31 STO 07
 32 RCL 05
 33 2
 34 *
 35 RCL 04
 36 P-R
 37 RCL 06
 38 +
 39 X<>Y
 40 RCL 07
 41 +
 42 X<>Y
 43 R-P
 44 X<>Y
 45 X<0?
 46 GTO 01
 47 X<>Y
 48 GTO 02
```

Input for Lens 1

Input for Lens 2

Calculate

```
 49◆LBL 01
 50 180
 51 +
 52 X<>Y
 53 CHS
 54◆LBL 02
 55 STO 08
 56 X<>Y
 57 2
 58 /
 59 STO 09
 60 RCL 01
 61 RCL 04
 62 +
 63 RCL 08
 64 -
 65 2
 66 /
 67 RCL 00
 68 +
 69 RCL 03
 70 +
 71 STO 10
 72 0
 73 RCL 08
 74 X<=Y?
 75 GTO 04
 76 CHS
 77 STO 08
 78 CHS
 79 RCL 20
 80 +
 81 STO 10
 82 90
 83 RCL 09
 84 X>Y?
 85 GTO 05
 86 +
 87 STO 09
 88 GTO 04
 89◆LBL 05
 90 X<>Y
 91 -
 92 STO 09
 93◆LBL 04
 94 FIX 2
 95 RCL 10
 96 STO 00
 97 CLA
 98 ARCL 10
 99 "⊢ SPH S
```

Output

Program Listings

```
UM"
100 AVIEW
101 STOP
102 RCL 08
103 STO 01
104 CLA
105 ARCL 08
106 "⊢ CYL S
UM"
107 AVIEW
108 STOP
109 FIX 0
110 RCL 09
111 STO 02
112 CLA
113 ARCL 09
114 "⊢ AX SU
M"
115 AVIEW
116 STOP
117 GTO 03
118 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	S_1	50	
	C_1		
	A_1		
	S_2		
	C_2		
05	A_2	55	
	Scratch		
	Scratch		
	Scratch		
	Scratch		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 40 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ACUITY DEMAND FROM LETTER SIZE AND WORKING DISTANCE

This program calculates acuity demand in Snellen form, Decimal notation, m notation, point type (small case letters) and point type (capital or large case letters) from letter size in millimeters and working distance in meters.

Remarks:

Point type numbers are approximations as there is not a direct relationship between letter size or acuity and point type.

Reference:

"A Guide to the Care of Low Vision Patients", Published by the American Optomeric Association, 1974.

Example:

What is the acuity demand of a 4.4 mm letter E if viewed from 40cm, the normal near working distance.

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 004		
[XEQ] [ALPHA] VISION [ALPHA]	LTR SIZE - MM	
4.4 [R/S]	WRK DIST - M	
.4 [R/S]	0.13 DEC. N.	(Decimal notation)
[R/S]	20. S. ACTY	
[R/S]	151. S. ACTY	(Snellen form = 20/151)
[R/S]	3.03 M. NOT.	(m notation)
[R/S]	18. PT. SML.	(pt. type small)
[R/S]	26. PT. LRG.	(pt. type large)

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] Vision	LTR SIZE - MM
3.	Key height of letter.	mm	[R/S]	WRK DIST - M
4.	Key working distance.	m	[R/S]	( ) DEC. N.
5.	Obtain Snell Acuity numerator.		[R/S]	( ) S. ACTY
6.	Obtain Snell Acuity denominator.		[R/S]	( ) S. ACTY
7.	Obtain m. notation.		[R/S]	( ) M. NOT.
8.	Obtain point type (small).		[R/S]	( ) PT. SML.
9.	Obtain point type (large).		[R/S]	( ) PT. LRG.
10.	For a new problem, press [R/S] and go to			
	step 3.		[R/S]	LTR SIZE - MM

Program Listings

```
01◆LBL "VIS
ION"
02◆LBL 02
03 "LTR SIZ
E-MM"
04 PROMPT
05 1 E3
06 /
07 STO 00
08 "WRK DIS
T-M"
09 PROMPT
10 STO 01
11 ADV
12 RCL 00
13 RCL 01
14 /
15 R-D
16 60
17 *
18 5
19 /
20 1/X
21 FIX 2
22 CLA
23 ARCL X
24 "⊢ DEC.
N."
25 AVIEW
26 STOP
27 STO 02
28 20
29 ENTER↑
30 RCL 02
31 /
32 STO 03
33 20
34 XEQ 01
35 RCL 03
36 XEQ 01
37 .4
38 RCL 02
39 /
40 FIX 2
41 CLA
42 ARCL X
43 "⊢ M. NO
T."
44 AVIEW
45 STOP
46 RCL 00
```

Input

Calculate Decimal Notation

Calculate Snellen Form

Calculate m notation

```
47 .0254
48 /
49 103
50 1/X
51 /
52 FIX 0
53 CLA
54 ARCL X
55 "⊢ PT. S
ML."
56 AVIEW
57 STOP
58 RCL 00
59 .0254
60 /
61 150
62 1/X
63 /
64 CLA
65 ARCL X
66 "⊢ PT. L
RG."
67 AVIEW
68 ADV
69 STOP
70 GTO 02
71◆LBL 01
72 FIX 0
73 CLA
74 ARCL X
75 "⊢ S. AC
TY"
76 AVIEW
77 STOP
78 RTN
79 .END.
```

Calculate point types

Output for Snellen form

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	LTR SIZE ÷ 1000	50	
	W, DISTANCE		
	ACUITY D.		
	SNELLEN		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 31 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONTACT LENS/TELESCOPE CALCULATIONS

This program will compute power of contact lens, P_{CL}; power of spectacle lens, P_{SP}; magnification of telescope system, M_{TS}; and vertex distance required, V_{DR} given any two of the four variables.

Equations:

$$M_{TS} = \frac{P_{CL}}{P_{SP}}$$

$$V_{DR} = \frac{1}{P_{SP}} - \frac{1}{P_{CL}}$$

where:

M_{TS} is Magnification of telescope system in diopters.

P_{CL} is Power of contact lens in diopters

P_{SP} is Power of spectacle lens in diopters

V_{DR} is vertex distance required in millimeters

Lens configeration.



Reference:

"Guide to Care of Low Vision Patients" published by the American Optometric Association, 1974 pp 120,121.

Example:

Given P_{CL} = -45 D. and P_{SP} = 25 D., find M_{TS} and V_{DR}.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] CLENS [ALPHA]	PCL?
45 [CHS] [R/S]	PSP?
25 [R/S]	MTS?
[R/S]	VDR?
[R/S]	MTS=1.80
[R/S]	VDR=17.78

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] CLENS	PCL?
3.	Key P_{CL} if known ([R/S] if unknown)	P_{CL}	[R/S]	PSP?
4.	Key P_{SP} if known ([R/S] if unknown)	P_{SP}	[R/S]	MTS?
5.	Key M_{TS} if known ([R/S] if unknown)	M_{TS}	[R/S]	VDR?
6.	Key V_{DR} if known ([R/S] if unknown)	V_{DR}	[R/S]	ANS 1
			[R/S]	ANS 2
7.	For a new problem, press [R/S] and go to		[R/S]	PCL?
	step 3.			

Program Listings

```
 01◆LBL "CLE
NS"
 02◆LBL 09
 03 CLX
 04 CLRG
 05 "PCL ?"
 06 PROMPT
 07 ABS
 08 STO 00
 09 CLX
 10 "PSP ?"
 11 PROMPT
 12 STO 01
 13 CLX
 14 "MTS ?"
 15 PROMPT
 16 STO 02
 17 CLX
 18 "VDR ?"
 19 PROMPT
 20 STO 03
 21 RCL 00
 22 X=0?
 23 GTO 01
 24 RCL 01
 25 X=0?
 26 GTO 02
 27 /
 28 XEQ 00
 29 RCL 01
 30 1/X
 31 RCL 00
 32 1/X
 33 -
 34 1 E3
 35 *
 36 XEQ 06
 37 GTO 09
 38◆LBL 01
 39 RCL 01
 40 X=0?
 41 GTO 03
 42 RCL 02
 43 X=0?
 44 GTO 04
 45 *
 46 XEQ 07
 47 RCL 01
 48 1/X
 49 X<>Y
 50 1/X
```

Input

Start computation

Calculate MTS and VDR from PCL and PSP

Calculate PSP and VDR from PCL and MTS

```
 51 -
 52 1 E3
 53 *
 54 XEQ 06
 55 GTO 09
 56◆LBL 02
 57 RCL 00
 58 RCL 02
 59 X=0?
 60 GTO 05
 61 /
 62 XEQ 08
 63 1/X
 64 RCL 00
 65 1/X
 66 -
 67 1 E3
 68 *
 69 XEQ 06
 70 GTO 09
 71◆LBL 03
 72 RCL 02
 73 RCL 03
 74 1 E3
 75 /
 76 *
 77 1/X
 78 RCL 03
 79 1 E3
 80 /
 81 1/X
 82 X<>Y
 83 -
 84 XEQ 08
 85 RCL 02
 86 *
 87 XEQ 07
 88 GTO 09
 89◆LBL 04
 90 RCL 03
 91 1 E3
 92 /
 93 RCL 01
 94 *
 95 1
 96 -
 97 RCL 01
 98 X<>Y
 99 /
100 XEQ 07
101 RCL 01
```

Calculate PCL and PSP from MTS and VDR

Calculate PCL and MTS from PSP and VDR

Calculate PCL and MTS from PSP and VDR

Program Listings

```
102 /
103 XEQ 00
104 GTO 09
105◆LBL 05
106 RCL 00
107 RCL 03
108 1 E3
109 /
110 RCL 00
111 *
112 1
113 +
114 RCL 00
115 X<>Y
116 /
117 XEQ 08
118 RCL 00
119 X<>Y
120 /
121 XEQ 00
122 GTO 09
123◆LBL 00
124 "MTS="
125 ARCL X
126 AVIEW
127 STOP
128 RTN
129◆LBL 06
130 "VDR="
131 ARCL X
132 AVIEW
133 STOP
134 RTN
135◆LBL 07
136 "PCL="
137 ARCL X
138 AVIEW
139 STOP
140 RTN
141◆LBL 08
142 "PSP="
143 ARCL X
144 AVIEW
145 STOP
146 RTN
147 .END.
```

Calculate PSP
and MTS from
PCL aad VDR

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	P_{CL}	50	
	P_{SP}		
	M_{TS}		
	V_{DR}		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 39 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

NEEDED MAGNIFICATION, ADD, AND WORKING DISTANCE

This program calculates how much magnification is needed to obtain a certain acuity and also how that magnification can be obtained. Given the patient, best visual acuity and the needed acuity, the program calculates needed magnification, lens to obtain that magnification and distance at which the lens is to be used. Then, by entering the magnification of a telescope to be used, the program calculates the new working distance and lens cap to be used.

Equations:

$$\text{MAG} = \frac{\text{BEST VA}}{\text{NEEDED VA}}$$

$$\text{LENS POWER} = \frac{\text{MAG}}{\text{DISTANCE}}$$

$$\text{WD} = \frac{1}{\text{LENS POWER}}$$

where:

VA is visual acuity, Enter denominator only. Example 20/40, enter 40 for VA

Reference:

"Low Vision Care" by Mehr and Fried.

Example:

Best V. Acty is 20/200
Needed VA is 20/50
Reference Distance is 0.4 m

What is the magnification, working distance, and lens power?

If a 2x telescope is used, what is the working distance and lens power?

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 005		
[XEQ] [ALPHA] NMAG [ALPHA]	BEST V. ACTY.?	
200 [R/S]	DESIRED V.A.?	
50 [R/S]	REF. DIST.?	
.4 [R/S]	MAG=4.00	
[R/S]	WD=0.10	(M)
[R/S]	PWR=10.00	(D)
[R/S]	TEL. MAG.?	
2 [R/S]	WD=0.20	(M)
[R/S]	PWR=5.00	(D)

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] NMAG	BEST V. ACTY?
3.	Key best visual acuity.	BEST	[R/S]	DESIRED V.A.?
4.	Key needed (desired) visual acuity.	NEEDED	[R/S]	REF. DIST.
5.	Key reference distance.	DIST (M)	[R/S]	MAG=
6.	Calculate working distance.		[R/S]	WD=
7.	Calculate lens power.		[R/S]	PWR=
8.	To enter telescope magnification:		[R/S]	TEL. MAG.?
9.	Key telescope magnification.	MAG	[R/S]	WD=
10.	Calculate lens power.		[R/S]	PWR=
11.	For a new problem, press [R/S] and go to			
	step 3.		[R/S]	BEST V? ACTY?

Program Listings

```
 01◆LBL "NMA
G"
 02 FIX 2
 03◆LBL 01
 04 "BEST V.
 ACTY.?"
 05 PROMPT
 06 STO 00
 07 CLX
 08 "DESIRED
 V.A. ?"
 09 PROMPT
 10 STO 01
 11 CLX
 12 "REF. DI
ST. ?"
 13 PROMPT
 14 STO 02
 15 RCL 00
 16 RCL 01
 17 /
 18 "MAG="
 19 ARCL X
 20 AVIEW
 21 STOP
 22 STO 03
 23 RCL 02
 24 RCL 03
 25 /
 26 "WD="
 27 ARCL X
 28 AVIEW
 29 STOP
 30 STO 04
 31 1/X
 32 "PWR="
 33 ARCL X
 34 AVIEW
 35 STOP
 36 "TEL. MA
G. ?"
 37 PROMPT
 38 RCL 04
 39 *
 40 "WD="
 41 ARCL X
 42 AVIEW
 43 STOP
 44 1/X
 45 "PWR="
 46 ARCL X
```

Input

Magnitude

Working distance

Power

Tel. Mag. Input

New WD and PWR

```
 47 AVIEW
 48 STOP
 49 GTO 01
 50 .END.
```

60 70 80 90 00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	X_1	50	
	X_2		
	Ref. Dist.		
	MAG.		
	W.D.		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 24 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EFFECTIVE, EQUIVALENT, AND NEUTRALIZING POWER

This program calculates the effective power, the equivalent power and the neutralizing power of a lens given the front and back surface power, the thickness, and the refractive index of the lens.

Equations:

$$\text{Effective power: } D_e = D_1 + D_2 + \frac{t}{n} D_1^2$$

$$\text{Equivalent power: } D_{eq} = D_1 + D_2 - \frac{t}{n} D_1 D_2$$

$$\text{Neutralizing power: } D_{ne} = D_1 + D_2 + \frac{t}{n} D_2^2$$

where:

D_1 = Front surface power

D_2 = Back surface power

t = Thickness

n = Index of refraction

Remarks:

Powers are expressed in diopters and thickness' are expressed in millimeters.

Reference:

Ophthalmic Mechanics and Dispensing by John P. Epting and Frank Morgret, Jr., Chilton Book Co., Radnor, PA, 1964.

Example:

Calculate the effective power, equivalent power, and neutralizing power of the following lens:

Front surface power = 16.00 Diopters
Back surface power = -6.00 Diopters
Thickness = 7.00 mm
Index of refraction (n) = 1.523

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] EPWR [ALPHA]	F. SUR. POWER?
16 [R/S]	B. SUR. POWER?
6 [CHS] [R/S]	THICKNESS?
7 [R/S]	INDEX OF R.?
1.523 [R/S]	EF=11.18
[R/S]	EQ=10.44
[R/S]	NT=10.17

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Start program.		[XEQ] EPWR	F. SUR.POWER?
3.	Input front surface power.	D_1 (D)	[R/S]	B. SUR.POWER?
4.	Input back surface power.	D_2 (D)	[R/S]	THICKNESS?
5.	Input thickness.	t (mm)	[R/S]	INDEX OF R.?
6.	Input index of refraction.	n	[R/S]	EF=
			[R/S]	EQ=
			[R/S]	NT=
7.	For a new problem, press [R/S] and go to			
	step 3.		[R/S]	F. SUR. POWER?

Program Listings

```
 01◆LBL "EPW
R"
 02◆LBL 01
 03 "F. SUR.
 POWER?"
 04 PROMPT
 05 ENTER↑
 06 STO 00
 07 ST* 00
 08 STO 02
 09 STO 04
 10 "B. SUR.
 POWER?"
 11 PROMPT
 12 ST+ 02
 13 ST* 04
 14 STO 01
 15 ST* 01
 16 " THICKN
ESS?"
 17 PROMPT
 18 1000
 19 /
 20 ENTER↑
 21 "INDEX O
F R. ?"
 22 PROMPT
 23 /
 24 STO 03
 25 RCL 02
 26 RCL 03
 27 RCL 00
 28 *
 29 +
 30 FIX 2
 31 "EF="
 32 ARCL X
 33 AVIEW
 34 STOP
 35 RCL 04
 36 RCL 03
 37 *
 38 CHS
 39 RCL 02
 40 +
 41 "EQ="
 42 ARCL X
 43 AVIEW
 44 STOP
 45 RCL 02
 46 RCL 03
 47 RCL 01
 48 *
 49 +
 50 "NT="
 51 ARCL X
 52 AVIEW
 53 STOP
 54 GTO 01
 55 .END.
```

Input D_1

Input D_2

Input t

Input n

Effective power

Equivalent power

Neutralizing power

60	
70	
80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	D_1^2	50	
	D_2^2		
	$D_1 + D_2$		
	t/1000n		
	D_1D_2		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 24 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

POSITIONAL EFFECTIVE POWER

This program calculates the effective power of a given lens at a new vertex distance given its power and original vertex distance.

Equations:

$$D = \frac{1000}{F \pm d}$$

Where:

D = Dioptric value of lens (plus or minus diopters).

F = Focal length of lens.

d = Distance between original and new lens position in millimeters.

Reference:

Opthalmic Mechanics and Dispensing, John P. Epting and Frank Morgret, Jr., Chilton Book Company, Radnor, PA, 1964.

Example:

A +10.00 lens corrects a hyperopic ametropia when placed at 14 mm. What is the effective power if it is moved further to 18 mm?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] LPWR [ALPHA]	LENS PWR. ?
10 [R/S]	ORIG. V. DIST.?
14 [R/S]	NEW V. DIST.?
18 [R/S]	10.42 DIOP.

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] LPWR	LENS PWR.?
3.	Input lens power in + or - diopters.	DIOPTERS	[R/S]	ORIG. V.DIST?
4.	Input original vertex distance.	mm	[R/S]	NEW V. DIST.?
5.	Input new vertex distance.	mm	[R/S]	ANSWER IN
				DIOPTERS
6.	For new problem,		[R/S]	LENS PWR.?
	Continue with step 3.			

Program Listings

```
 01◆LBL "LPW
R"
 02◆LBL 00
 03 "LENS PW
R. ?"
 04 PROMPT
 05 1 E3
 06 X<>Y
 07 /
 08 STO 00
 09 "ORIG. V
. DIST.?"
 10 PROMPT
 11 STO 01
 12 "NEW V.
DIST.?"
 13 PROMPT
 14 STO 02
 15 RCL 00
 16 RCL 02
 17 RCL 01
 18 -
 19 X>0?
 20 GTO 02
 21 CHS
 22 +
 23 1/X
 24 GTO 04
 25◆LBL 02
 26 -
 27 1/X
 28◆LBL 04
 29 1 E3
 30 *
 31 CLA
 32 ARCL X
 33 "⊢ DIOP.
"
 34 AVIEW
 35 STOP
 36 GTO 00
 37 .END.
```

Input lens power

Input original distance

Input new distance

Calculate

Output power

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	F	50	
	V1		
	V2		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 17 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PRATT, SHEARD, AND PERCIVAL METHODS OF NEAR Rx

This program calculates near Rx (.4) by Pratt's Sheard's, and Percival's criterion. Answers are given in diopters and prism diopters.

Equations:

Pratt: $$D.\ ADD = \frac{\text{phoria + acc. lag (14B)}}{AC/A + 1}$$

$$\Delta\ ADD = \frac{\text{phoria + acc. lag (14B)}}{CA/C + 1}$$

Sheard: $\Delta = 2$ DEMAND RESERVE/3

$D = \Delta/AC/A$

Percival: $\Delta = 2$ LESSER GREATER/3

$D = \Delta/AC/A$

Remarks:

Phorias Exo entered as minus.

Example:

Use the following data to calculate NEAR Rx (.4m) by Pratt's, Sheard's, and Percival's criterion.

7A = 0 Diopters	PD = 60 mm
13B = 3 eso Prism Diopters	d = 0.4 M
14B = 1.25 Diopters	BO = 15Δ Prism Diopters
15B = 2 exo Prism Diopters	B1 = 3Δ Prism Diopters

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] NRX [ALPHA]	7A?
0 [R/S]	13B?
3 [R/S]	14B?
1.25 [R/S]	15B?
2 [CHS] [R/S]	PD?
60 [R/S]	d?
.4 [R/S]	BO?
15 [R/S]	B1?
3 [R/S]	PRATT
	DELTA 1.05
[R/S]	DELTA BO 7.00
[R/S]	SHEARD
	DELTA 0.25
[R/S]	DELTA BO 1.00
[R/S]	PERCIVAL
	DELTA 0.75
[R/S]	DELTA BO 3.00

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] NRX	7A?
3.	Input 7A.	DIOPTERS	[R/S]	13B?
4.	Input 13B (if Exo minus (-)).	PRISM DIOPTERS	[R/S]	14B?
5.	Input 14B.	DIOPTERS	[R/S]	15B?
6.	Input 15B (if Exo minus (-)).	PRISM DIOPTERS	[R/S]	PD?
7.	Input PD in mm.	mm	[R/S]	d?
8.	Input d in m.	m	[R/S]	BO?
9.	Input BO Duction.	PRISM DIOPTERS	[R/S]	B1?
10.	Input B1 Duction.	PRISM DIOPTERS	[R/S]	
				PRATT
			[R/S]	DELTA ( )
			[R/S]	DELTA BO ( )
			[R/S]	SHEARD
				DELTA ( )
			[R/S]	DELTA BO ( )
			[R/S]	PERCIVAL
				DELTA ( )
			[R/S]	DELTA BO ( )

Program Listings

```
01◆LBL "NRX
"
02 "7A?"
03 PROMPT
04 STO 00
05 "13B?"
06 PROMPT
07 STO 02
08 "14B?"
09 PROMPT
10 STO 01
11 "15B?"
12 PROMPT
13 STO 10
14 "PD?"
15 PROMPT
16 10
17 /
18 STO 03
19 "d?"
20 PROMPT
21 1/X
22 STO 04
23 "B0?"
24 PROMPT
25 STO 06
26 "B1?"
27 PROMPT
28 "PRATT"
29 AVIEW
30 STO 07
31 RCL 02
32 RCL 03
33 /
34 RCL 01
35 RCL 00
36 -
37 +
38 STO 05
39 RCL 02
40 RCL 10
41 -
42 RCL 03
43 /
44 RCL 01
45 RCL 00
46 -
47 /
48 STO 08
49 RCL 08
50 1
```

Input

Pratt Analysis

```
51 +
52 RCL 05
53 X<>Y
54 /
55 "DELTA "
56 ARCL X
57 AVIEW
58 STOP
59 RCL 01
60 RCL 00
61 -
62 RCL 04
63 X<>Y
64 -
65 RCL 04
66 /
67 1
68 +
69 RCL 05
70 X<>Y
71 /
72 6
73 *
74 "DELTA B
0 "
75 ARCL X
76 AVIEW
77 STOP
78 SF 05
79 "SHEARD"
80 AVIEW
81 RCL 02
82 X>0?
83 GTO 01
84 2
85 *
86 ABS
87 RCL 06
88 X<>Y
89 X<=Y?
90 GTO 02
91 RCL 02
92 2
93 *
94 RCL 06
95 +
96 3
97 /
98 ENTER↑
99 XEQ 03
100 GTO 07
```

Sheard Analysis

Program Listings

```
101◆LBL 02
102 XEQ 09
103 GTO 07
104◆LBL 01
105 2
106 *
107 RCL 07
108 X>Y?
109 GTO 02
110 RCL 02
111 ABS
112 2
113 *
114 RCL 07
115 -
116 3
117 /
118 ENTER↑
119 XEQ 03
120◆LBL 07
121 RCL 06
122 RCL 07
123 +
124 3
125 /
126 RCL 06
127 X>Y?
128 GTO 05
129 -
130 CHS
131 ENTER↑
132 GTO 03
133◆LBL 05
134 RCL 07
135 RCL 06
136 +
137 3
138 /
139 RCL 07
140 X>Y?
141 GTO 06
142 -
143 ENTER↑
144 GTO 03
145◆LBL 06
146 RCL 06
147 RCL 07
148 +
149 3
150 /
151 RCL 06
```

Percival Analysis

```
152 X>Y?
153 GTO 09
154 -
155 ENTER↑
156◆LBL 03
157 RCL 08
158 RCL 03
159 *
160 /
161 "DELTA "
162 ARCL X
163 AVIEW
164 STOP
165 X<>Y
166 "DELTA B
0 "
167 ARCL X
168 AVIEW
169 STOP
170 FC?C 05
171 RTN
172 "PERCIVA
L"
173 AVIEW
174 RTN
175◆LBL 09
176 0
177 "DELTA "
178 ARCL X
179 AVIEW
180 STOP
181 "DELTA B
0 "
182 ARCL X
183 AVIEW
184 STOP
185 RTN
186 .END.
```

Output for special case

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	7A	50	
	14B		
	13B		
	PD		
	WD		
05	Scratch	55	
	B0 Duction		
	B1 Duction		
	Scratch		
	15B		
10	Scratch	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 55 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
05		1st use of LBL 03	2nd use of LBL 03

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FOUR ACCOMMODATIVE Rx's AND THEIR AVERAGE

This program calculates the near Rx using four different accommodative methods.

Equations:

(1) Near Point of Accommodative Method (NPA)

$$ADD = 1/d\ NPA + P/2$$

(2) Positive Relative Amplitude Method (PRA)

$$ADD = 1/d - \Delta(20-P) + 1/d/2$$

(3) Midrange of Positive and Negative Relative Amplitude Method (MID)

$$ADD = \left(21 - \frac{\Delta(20-21)}{2}\right) - P$$

(4) Normative Lag Method for Binocular Cross - Cylinder (NORM)

$$ADD = (14B - P) - .75$$

Example:

Determine the near Rx using four accommodative methods from the following data:

"P" = 0 Diopters
d = 0.4 Meters
NPA = 0.333 Meters
20 = -1.00 Diopters
21 = 2.75 Diopters
14B = 1.25 Diopters

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] ARX [ALPHA]	P?
0 [R/S]	d?
.4 [R/S]	NPA?
.333 [R/S]	20?
1 [CHS] [R/S]	21?
2.75 [R/S]	14B?
1.25 [R/S]	NPA=0.9985
[R/S]	PRA=0.7500
[R/S]	MID=0.8750
[R/S]	NORM=0.5000
[R/S]	AVG=0.7809

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] ARX	P?
3.	Input "P" value.	P(diopters)	[R/S]	d?
4.	Input working distance, d.	d(meters)	[R/S]	NPA?
5.	Input NPA.	NPA(meters)	[R/S]	20?
6.	Input #20 Positive Relative Acc.	#20(diopters)	[R/S]	21?
7.	Input #21 Negative Relative Acc.	#21(diopters)	[R/S]	14B?
8.	Input #14B Binocular Cross - Cylinder	#14B(diopters)	[R/S]	NPA=
	Outputs are in diopters.		[R/S]	PRA=
			[R/S]	MID=
			[R/S]	NORM=
			[R/S]	AVE=

Program Listings

```
 01◆LBL "ARX
"
 02◆LBL 04
 03 "P?"
 04 PROMPT
 05 STO 00
 06 "d?"
 07 PROMPT
 08 1/X
 09 STO 01
 10 "NPA?"
 11 PROMPT
 12 X≠0?
 13 1/X
 14 STO 02
 15 "20?"
 16 PROMPT
 17 STO 03
 18 "21?"
 19 PROMPT
 20 STO 04
 21 "14B?"
 22 PROMPT
 23 STO 05
 24 RCL 00
 25 RCL 02
 26 X=0?
 27 GTO 03
 28 +
 29 2
 30 /
 31 RCL 01
 32 X<>Y
 33 -
 34 X<0?
 35 0
 36 "NPA="
 37 ARCL X
 38 AVIEW
 39 STOP
 40 STO 06
 41◆LBL 02
 42 RCL 03
 43 RCL 00
 44 -
 45 ABS
 46 RCL 01
 47 +
 48 2
 49 /
 50 RCL 01
```

Input

NPA
calculation

PRA
calculation

```
 51 X<>Y
 52 -
 53 X<0?
 54 0
 55 "PRA="
 56 ARCL X
 57 AVIEW
 58 STOP
 59 RCL 06
 60 +
 61 STO 06
 62 RCL 03
 63 RCL 04
 64 -
 65 ABS
 66 2
 67 /
 68 RCL 04
 69 X<>Y
 70 -
 71 RCL 00
 72 -
 73 X<0?
 74 0
 75 "MID="
 76 ARCL X
 77 AVIEW
 78 STOP
 79 RCL 06
 80 +
 81 STO 06
 82 RCL 05
 83 RCL 00
 84 -
 85 .75
 86 -
 87 X<0?
 88 0
 89 "NORM="
 90 ARCL X
 91 AVIEW
 92 STOP
 93 RCL 06
 94 +
 95 4
 96 /
 97 "AVG="
 98 ARCL X
 99 AVIEW
100 STOP
101 GTO 04
```

Mid. calculation

Norm. calculation

Avg. calculation

Program Listings

```
102◆LBL 03
103 0
104 "NPA="
105 ARCL X
106 AVIEW
107 STOP
108 GTO 02
109 .END.
```

Special case

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	P	50	
	d		
	NPA		
	20		
	21		
05	14B	55	
	SUM		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 31 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

NOTES

NOTES

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Optometry I (General)

OPTOMETRY I (GENERAL)

ANISEIKONIA

PROGRAM REGISTERS NEEDED: 98

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 7)

ROW 2 (7 - 8)

ROW 3 (8 - 11)

ROW 4 (11 - 15)

ROW 5 (16 - 28)

ROW 6 (29 - 39)

ROW 7 (39 - 47)

ROW 8 (48 - 57)

ROW 9 (58 - 67)

ROW 10 (68 - 78)

ROW 11 (79 - 87)

ROW 12 (88 - 98)

ROW 13 (98 - 104)

ROW 14 (104 - 111)

ROW 15 (112 - 117)

ROW 16 (117 - 121)

ROW 17 (122 - 127)

ROW 18 (127 - 132)

ROW 19 (133 – 137)

ROW 20 (137 – 140)

ROW 21 (140 – 145)

ROW 22 (146 – 152)

ROW 23 (152 – 155)

ROW 24 (156 – 164)

ROW 25 (164 – 166)

ROW 26 (167 – 172)

ROW 27 (173 – 178)

ROW 28 (178 – 182)

ROW 29 (182 – 184)

ROW 30 (185 – 188)

ROW 31 (189 – 196)

ROW 32 (197 – 206)

ROW 33 (207 – 217)

ROW 34 (218 – 226)

ROW 35 (227 – 238)

ROW 36 (239 – 246)

ROW 37 (247 - 254)

ROW 38 (255 - 266)

ROW 39 (267 - 274)

ROW 40 (275 - 283)

ROW 41 (284 - 291)

ROW 42 (292 - 302)

ROW 43 (303 - 310)

ROW 44 (310 - 320)

ROW 45 (321 - 329)

ROW 46 (329 - 336)

ROW 47 (337 - 347)

ROW 48 (348 - 356)

ROW 49 (357 - 365)

ROW 50 (366 - 376)

ROW 51 (377 - 386)

ROW 52 (387 - 396)

ROW 53 (397 - 399)

CROSSED PRISM RESULTANT

PROGRAM REGISTERS NEEDED: 15

ROW 1 (1 - 3)

ROW 2 (3 - 8)

ROW 3 (8 - 11)

ROW 4 (12 - 24)

ROW 5 (25 - 34)

ROW 6 (34 - 38)

ROW 7 (39 - 49)

ROW 8 (49 - 53)

OBLIQUE CYLINDER SUMS

PROGRAM REGISTERS NEEDED: 30

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 – 4)

ROW 2 (4 – 7)

ROW 3 (7 – 13)

ROW 4 (14 – 17)

ROW 5 (17 – 21)

ROW 6 (22 – 34)

ROW 7 (35 – 46)

ROW 8 (47 – 56)

ROW 9 (57 – 69)

ROW 10 (70 – 80)

ROW 11 (81 – 90)

ROW 12 (91 – 99)

ROW 13 (99 – 105)

ROW 14 (105 – 108)

ROW 15 (109 – 114)

ROW 16 (114 – 118)

ACUITY DEMAND FROM LETTER SIZE
AND WORKING DISTANCE
PROGRAM REGISTERS NEEDED: 28

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 3)

ROW 2 (3 - 5)

ROW 3 (5 - 8)

ROW 4 (8 - 19)

ROW 5 (20 - 24)

ROW 6 (24 - 33)

ROW 7 (33 - 40)

ROW 8 (40 - 43)

ROW 9 (43 - 49)

ROW 10 (50 - 55)

ROW 11 (55 - 59)

ROW 12 (60 - 66)

ROW 13 (66 - 71)

ROW 14 (72 - 75)

ROW 15 (75 - 79)

ROW 1 (1 - 5)

ROW 2 (5 - 10)

ROW 3 (10 - 16)

ROW 4 (17 - 23)

ROW 5 (24 - 33)

ROW 6 (34 - 41)

ROW 7 (41 - 50)

ROW 8 (51 - 58)

ROW 9 (59 - 67)

ROW 10 (67 - 74)

ROW 11 (75 - 84)

ROW 12 (84 - 91)

ROW 13 (92 - 102)

ROW 14 (103 - 110)

ROW 15 (111 - 121)

ROW 16 (121 - 126)

ROW 17 (127 - 134)

ROW 18 (135 - 142)

ROW 19 (142 - 147)

NEEDED MAGNIFICATION--ADD--AND
WORKING DISTANCE
PROGRAM REGISTERS NEEDED: 20

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 4)

ROW 2 (4 - 4)

ROW 3 (5 - 8)

ROW 4 (8 - 12)

ROW 5 (12 - 17)

ROW 6 (18 - 25)

ROW 7 (26 - 32)

ROW 8 (32 - 36)

ROW 9 (36 - 41)

ROW 10 (41 - 48)

ROW 11 (49 - 50)

EFFECTIVE--EQUIVALENT--
AND NEUTRALIZING POWER
PROGRAM REGISTERS NEEDED: 20

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 3)

ROW 2 (3 - 5)

ROW 3 (6 - 10)

ROW 4 (10 - 14)

ROW 5 (15 - 16)

ROW 6 (16 - 21)

ROW 7 (21 - 25)

ROW 8 (26 - 33)

ROW 9 (34 - 42)

ROW 10 (43 - 51)

ROW 11 (52 - 55)

POSITIONAL EFFECTIVE POWER

PROGRAM REGISTERS NEEDED: 15

ROW 1 (1 - 3)

ROW 2 (3 - 6)

ROW 3 (7 - 9)

ROW 4 (9 - 12)

ROW 5 (12 - 17)

ROW 6 (18 - 28)

ROW 7 (29 - 33)

ROW 8 (33 - 37)

PRATT--SHEARD--AND PERCIVAL
METHODS OF NEAR Rx
PROGRAM REGISTERS NEEDED: 45

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 4)

ROW 2 (5 - 9)

ROW 3 (10 - 15)

ROW 4 (16 - 23)

ROW 5 (23 - 28)

ROW 6 (28 - 40)

ROW 7 (41 - 53)

ROW 8 (54 - 59)

ROW 9 (60 - 72)

ROW 10 (73 - 75)

ROW 11 (76 - 81)

ROW 12 (82 - 92)

ROW 13 (93 - 102)

ROW 14 (102 - 111)

ROW 15 (112 - 122)

ROW 16 (123 - 133)

ROW 17 (134 - 144)

ROW 18 (145 - 156)

ROW 19 (157 - 162)

ROW 20 (163 - 166)

ROW 21 (167 - 172)

ROW 22 (172 - 177)

ROW 23 (177 - 181)

ROW 24 (181 - 186)

FOUR ACCOMMODATIVE Rx'S
AND THEIR AVERAGE
PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
OPTOMETRY I (GENERAL)

ROW 1 (1 - 5)

ROW 2 (6 - 12)

ROW 3 (13 - 19)

ROW 4 (20 - 27)

ROW 5 (28 - 36)

ROW 6 (37 - 48)

ROW 7 (49 - 56)

ROW 8 (57 - 69)

ROW 9 (70 - 77)

ROW 10 (78 - 88)

ROW 11 (89 - 95)

ROW 12 (96 - 102)

ROW 13 (103 - 109)

ROW 14 (109 - 109)

NOTES

NOTES

NOTES


HEWLETT
PACKARD

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

OPTOMETRY I

ANISEIKONIA
CROSSED PRISM RESULTANT
OBLIQUE CYLINDER SUMS
ACUITY DEMAND FROM LETTER SIZE AND WORKING DISTANCE
CONTACT LENS/TELESCOPE CALCULATIONS
NEEDED MAGNIFICATION, ADD, AND WORKING DISTANCE
EFFECTIVE, EQUIVALENT, AND NEUTRALIZING POWER
POSITIONAL EFFECTIVE POWER
PRATT, SHEARD, AND PERCIVAL METHODS OF NEAR RX
FOUR ACCOMMODATIVE RX'S AND THEIR AVERAGE

00041-90143 Rev. B 11/80 Printed in U.S.A.